



Exemplo: - Acidentes de Trânsito

Os dados apresentados a seguir correspondem a 68694 passageiros de automóveis e pequenos caminhões envolvidos no estado de Maine em 1991. A tabela classifica os passageiros por :

G – gênero (m – masculino e f- feminino)

L – localização do acidente (u –urbana e r –rural)

S – uso de cinto de segurança (s-sim e n-não)

I – ferimento (s-sim e n-não)

Gênero	Local	Cinto de Segurança	Ferimentos		Proporção de Passageiros.c/ Ferimentos
			Não	Sim	
Feminino	Urbana	Não	7287	996	0,12
		Sim	11587	759	0,06
	Rural	Não	3246	973	0,23
		Sim	6134	757	0,11
Masculino	Urbana	Não	10381	812	0,07
		Sim	10969	380	0,03
	Rural	Não	6123	1084	0,15
		Sim	6693	513	0,07

```
data acidente;
input genero $ local $ cintoseg $ dano $ freq;
datalines;
f u n n 7287
f u n s 996
f u s n 11587
f u s s 759
f r n n 3246
f r n s 973
f r s n 6134
f r s s 757
m u n n 10381
m u n s 812
m u s n 10969
m u s s 380
m r n n 6123
m r n s 1084
m r s n 6693
m r s s 513
;
run;

/* ##### MODELOS LOGLINEARES ##### */
/* Modelo Saturado */
proc catmod data=acidente;
title ' Modelo Saturado ' ;
weight freq;
model genero*local*cintoseg*dano=_response_ / noiter noparm noresponse noprofile;
loglin genero|local|cintoseg|dano;
run;
```

```

/* Modelo com Interações Triplas */
proc catmod data=acidente;
  title ' Modelo de Interação Tripla - (GLS,GLI,GSI,LSI) ' ;
  weight freq;
  model genero*local*cintoseg*dano=_response_ / noiter noparm noresponse noprofile;
  loglin genero|local|cintoseg genero|local|dano genero|cintoseg|dano local|cintoseg|dano ;
run;

/* Modelo com Interações Duplas */
proc catmod data=acidente;
  title ' Modelo de Interação Dupla - (GL,GS,LS,GI,LI,SI) ' ;
  weight freq;
  model genero*local*cintoseg*dano=_response_ / noiter noparm noresponse noprofile;
  loglin genero|local genero|cintoseg genero|dano local|cintoseg local|dano cintoseg|dano;
run;

/* Modelo de Independência Mútua */
proc catmod data=acidente;
  title ' Modelo de Independência Mútua - (G,L,S,I ) ' ;
  weight freq;
  model genero*local*cintoseg*dano=_response_ / noiter noparm noresponse noprofile;
  loglin genero local cintoseg dano;
run;

/* ##### */
/* Modelos com Interações Duplas e uma Interação Tripla */

/* Incluindo a interação GLS - Modelo (GLS,GI,LI,SI) */
proc catmod data=acidente;
  title ' Modelo (GLS,GI,LI,SI) ' ;
  weight freq;
  model genero*local*cintoseg*dano=_response_ / noiter noparm noresponse noprofile;
  loglin genero|local|cintoseg genero|dano local|dano cintoseg|dano ;
run;

/* Incluindo a interação GLI - Modelo (GLI,GS,LS,SI) */
proc catmod data=acidente;
  title ' Modelo (GLI,GS,LS,SI) ' ;
  weight freq;
  model genero*local*cintoseg*dano=_response_ / noiter noparm noresponse noprofile;
  loglin genero|local|dano genero|cintoseg local|cintoseg cintoseg|dano ;
run;

/* Incluindo a interação GSI - Modelo (GSI,GL,LS,LI) */
proc catmod data=acidente;
  title ' Modelo (GSI,GL,LS,LI) ' ;
  weight freq;
  model genero*local*cintoseg*dano=_response_ / noiter noparm noresponse noprofile;
  loglin genero|cintoseg|dano genero|local local|cintoseg local|dano ;
run;

/* Incluindo a interação LSI - Modelo (LSI,GL,GS,GI) */
proc catmod data=acidente;
  title ' Modelo (LSI,GL,GS,GI) ' ;
  weight freq;
  model genero*local*cintoseg*dano=_response_ / noiter noparm noresponse noprofile;
  loglin local|cintoseg|dano genero|local genero|cintoseg genero|dano ;
run;

```

```

/* ##### */

/* ***** Determinando as Estimativas dos Parâmetros ***** */
/* ***** Modelo (GLS,GI,LI,SI) * ***** */

proc catmod order=data;
  title 'Modelo (GLS,GI,LI,SI) ' ;
  weight freq;
  model genero*local*cintoseg*dano=_response_ / noprofile noresponse noiter ;
  loglin genero|local|dano genero|cintoseg local|cintoseg cintoseg|dano ;
run;

```

1. Identifique os modelos que se ajustam:

Para investigar a complexidade de modelos necessários, vamos considerar os modelos:

- *(G, L, S, I) contendo os termos com um único fator;*
- *(GL, GS, GI, LS, LI, SI) contendo os termos com um único fator e todos os termos com dois fatores*
- *(GLS, GLI, GSI, LSI) contendo os termos com um fator, com dois fatores e todos os termos com três fatores*

Deseja-se testar as seguintes hipóteses:

- H₀) O modelo testado se ajusta bem aos dados.
- H₁) O modelo testado não se ajusta aos dados.

	Modelo	G ²	Gl	p-value
1	(G, L, S, I)			
2	(GL, GS, GI, LS, LI, SI)			
3	(GLS, GLI, GSI, LSI)	1.33		

2. Identifique qual modelos se ajustam e o que melhor se ajusta:

	Modelo	G ²	Gl	p-value	Diferença	Gl	P-value
2	(GI, GL, GS, IL, IS, LS)	23.35	5	0.0003			
a	(GIL, GS, IS, LS)						
b	(GIS, GL, LS, IL)						
c	(GLS, GI, IL, IS)						
d	(ILS, GI, GL, GS)						
3	(GIL, GIS, GLS, ILS)	1.33	1	0.2496			

3.